Eigenvalues

Eigenvalues problem applications

Consider the following eigenvalue problem:

$$\phi''[s] + \lambda\,\phi[s] = 0,\quad 0 < s < 1$$

$$\text{BC1:}\quad \phi[0] = 0$$

$$\text{BC2:}\quad \phi'[1] + \alpha\,\phi[1] = 0$$

Use *Mathematica* to determine the eigenvalues and eigen-functions when α=0.1

In this example the eigenvalues cannot be found analytically. As before we tray determine the general solution

```
genSol = First[DSolve[ϕ''[s] + λ ϕ[s] == 0, ϕ, s]]
```

```
{ϕ → Function[{s}, C[1] Cos[s √λ] + C[2] Sin[s √λ]]}
```

Next, we determine the boundary conditions in terms of the general solution

```
BC1 = (ϕ[0] == 0) /. genSol
```

```
C[1] == 0
```

```
BC2 = (ϕ'[1] + α ϕ[1] == 0) /. genSol
```

$$\sqrt{\lambda}\, C[2] \,\text{Cos}\left[\sqrt{\lambda}\right] - \sqrt{\lambda}\, C[1] \,\text{Sin}\left[\sqrt{\lambda}\right] + \alpha \left(C[1] \,\text{Cos}\left[\sqrt{\lambda}\right] + C[2] \,\text{Sin}\left[\sqrt{\lambda}\right]\right) == 0$$

We could simplify BC2 by setting C[2]=0. However we will leave this step up to *Mathematica*. The coefficient matrix is

```
A = Map[Coefficient[First[#], {C[1], C[2]}] &, {BC1, BC2}]
```

$$\left\{\{1, 0\}, \left\{\alpha \,\text{Cos}\left[\sqrt{\lambda}\right] - \sqrt{\lambda}\,\text{Sin}\left[\sqrt{\lambda}\right], \sqrt{\lambda}\,\text{Cos}\left[\sqrt{\lambda}\right] + \alpha \,\text{Sin}\left[\sqrt{\lambda}\right]\right\}\right\}$$

The characteristic equation for the eigenvalues is found by setting the determinant equal to zero.

```
charEqn = (Det[A] // Simplify) == 0
```

$$\sqrt{\lambda}\,\text{Cos}\left[\sqrt{\lambda}\right] + \alpha \,\text{Sin}\left[\sqrt{\lambda}\right] == 0$$

The roots to this equation clearly depend on the parameter α. Thus our goal will be to determine the roots as a function of α. Here is a plot of the LHS for a particular value of α

```
Plot[First[charEqn] /. α → 0.1`, {λ, 0, 300}]
```



From the general solution we can deduce the $\lambda=0$ gives the trivial solution, unless of course $\alpha=0$. (We then have the problem given in Example 1). Thus for smal values of α the roots will be close to those found in Example 1. These are

$$\lambda_{n\_} := (2\,n+1)^2 \frac{\pi^2}{4} \;/;\; n > 0$$

$$\lambda_{n\_} := 0 \;/;\; n == 0$$

```
approxRoots = Table[λn, {n, 1, 5}]
```

$$\left\{\frac{9\,\pi^2}{4}, \frac{25\,\pi^2}{4}, \frac{49\,\pi^2}{4}, \frac{81\,\pi^2}{4}, \frac{121\,\pi^2}{4}\right\}$$

We can us these values as our initial guess for FindRoot

```
λvalues =
 Flatten[Map[FindRoot[Evaluate[(charEqn /. α → .1)], {λ, #}] &,
   approxRoots]]
```

```
{λ → 22.4061, λ → 61.8849, λ → 121.103, λ → 200.059, λ → 298.755}
```

Finally, we can assess whether we have all the roots for the give range of λ by plotting the data onto the original plot of the characteristic equation.

```
plt1 = Plot[First[charEqn] /. α → 0.1`, {λ, 0, 300},
   DisplayFunction → Identity];
Show[plt1,
 Graphics[
  Flatten[{PointSize[0.02`], RGBColor[0, 0, 1],
    (Point[{λ, 0}] /. #1 &) /@ λvalues}]],
 DisplayFunction → $DisplayFunction]
```



It is clear from the plot that we have all the roots in the selected range of λ . Our final task is to compute the eigen-functions. Since C[2] is zero the eigenvalues and eigenfunc-tions are

```
βp_ := λ /. λvalues[[p]]
```

```
Table[βn, {n, 1, 5}]
```

```
{22.4061, 61.8849, 121.103, 200.059, 298.755}
```

```
ϕn_[s_] := Sin[√βn s]
```

Here is a listing of the first five eigenfunctions

```
Table[ϕn[s], {n, 1, 5}]
```

```
{Sin[4.73351 s], Sin[7.86669 s],
 Sin[11.0047 s], Sin[14.1442 s], Sin[17.2845 s]}
```

We can readily show these eigenfunctions are orthogonal to each other

```
Chop[Map[∫_0^1 ϕ3[s] ϕ#[s] ⅆs &, Range[1, 5]]]
```

```
{0, 0, 0.500413, 0, 0}
```